*************************************************************************
*Table A5: Robots and incoming students: In-state and out-state students*
*************************************************************************

*Education
use "$final_data_outcomes/czone_Students.dta", clear

*Migration
merge 1:1 czone year using "$final_data_outcomes/czone_StudentMigrantsState.dta", assert(1 3) nogenerate

*Automation
merge 1:1 czone year using "$final_data_automation/czones_ExposureAutomation.dta", assert(1 3) nogenerate

*Imports
merge 1:1 czone year using "$final_data_imports/czone_ExposureChina", assert(1 3) nogenerate

*Covariates
merge m:1 czone using "$final_data_covariates/czone1990_Covariates.dta", assert(3) nogenerate //beginning of period
merge m:1 czone using "$final_data_covariates/IPEDS_institutionControlsCZ.dta", keep(1 3) nogenerate

xtset czone year

*Keep only periods of interest
keep if year<=2000

foreach var in public private uniTop30Dummy largest150 anyAid {
replace `var' = 0 if mi(`var')
}

eststo clear
eststo: ivreghdfe d_mig_inStStudentsN4 (expof_us_adj d_exp_us_stacked = expof_euro7_qo d_exp_iv_stacked) d_sc_70_90 $occupations_1990 $demographics_1990 $industry_shares_1990 $institutions_1990 [w=ipums_pop_1990], absorb(division##year statefip) cluster(statefip)
eststo: ivreghdfe d_mig_outStStudentsN4 (expof_us_adj d_exp_us_stacked = expof_euro7_qo d_exp_iv_stacked) d_sc_70_90 $occupations_1990 $demographics_1990 $industry_shares_1990 $institutions_1990 [w=ipums_pop_1990], absorb(division##year statefip) cluster(statefip)

estout using "$tables_appendix/Table A5.tex", style(tex) ///
varlabels(expof_us_adj "US robot exposure") ///
cells(b(fmt(%9.3f) star) se(par)) nolabel replace mlabels(none) collabels(none) starlevels(\$^{*}\$ .1 \$^{**}\$ .05 \$^{***}\$ .01) /// 
keep(expof_us_adj) ///
order(expof_us_adj)
